

Airy'n differentiaaliyhtälö

Airy'n differentiaaliyhtälö on hyvin yksinkertainen toisen kertaluvun lineaarinen ja homogeeninen differentiaaliyhtälö, joka kuitenkin ei ole ratkaistavissa tavallisten alkeisfunktioiden avulla:

```
airyyht = y''[x] - x y[x] == 0
-x y[x] + y''[x] == 0

ylrtk = DSolve[airyyht, y, x]
{{y -> Function[{x}, AiryAi[x] C[1] + AiryBi[x] C[2]]}}
```

Mathematica tuntee kuitenkin laajemman kokoelman funktioita, ja näiden avulla voidaan lausua sekä differentiaaliyhtälön yleinen ratkaisu että sen derivaatta:

```
airy = y /. First[ylrtk]
Function[{x}, AiryAi[x] C[1] + AiryBi[x] C[2]]

airy[x]
AiryAi[x] C[1] + AiryBi[x] C[2]

airyder = airy'
Function[{x}, AiryAiPrime[x] C[1] + AiryBiPrime[x] C[2]]

airyder[x]
AiryAiPrime[x] C[1] + AiryBiPrime[x] C[2]
```

Kaksi lineaarisesti riippumatonta yksittäisratkaisua saadaan antamalla sopivat alkuehdot ja ratkaisemalla näistä vakiot:

```
vakiot1 = Solve[{airy[0] == 1, airyder[0] == 0}, {C[1], C[2]}]
{{C[1] -> 1/2 3^(2/3) Gamma[2/3], C[2] -> 1/2 3^(1/6) Gamma[2/3]}}

vakiot2 = Solve[{airy[0] == 0, airyder[0] == 1}, {C[1], C[2]}]
{{C[1] -> -1/2 3^(1/3) Gamma[1/3], C[2] -> Gamma[1/3]/(2 3^(1/6))}}
```

Saadut lausekkeet sisältävät uuden erikoisfunktion, gammafunktion. Tälle käytetään yleensä symbolia Γ (kreikkalainen kirjain iso gamma).

Vakioita vastaavat yksittäisratkaisut ovat

```
rtk1 = airy[x] /. First[vakiot1]
```

$$\frac{1}{2} 3^{2/3} \text{AiryAi}[x] \text{Gamma}\left[\frac{2}{3}\right] + \frac{1}{2} 3^{1/6} \text{AiryBi}[x] \text{Gamma}\left[\frac{2}{3}\right]$$

```
rtk2 = airy[x] /. First[vakiot2]
```

$$-\frac{1}{2} 3^{1/3} \text{AiryAi}[x] \text{Gamma}\left[\frac{1}{3}\right] + \frac{\text{AiryBi}[x] \text{Gamma}\left[\frac{1}{3}\right]}{2 3^{1/6}}$$

Nämä voidaan — hieman Mathematican versiosta riippuen — saada myös suoraan DSolve-komennolla:

```
DSolve[{airyyht, y[0] == 1, y'[0] == 0}, y[x], x]
```


$$\left\{ \left\{ y[x] \rightarrow \frac{1}{2} \left(3^{2/3} \text{AiryAi}[x] \text{Gamma}\left[\frac{2}{3}\right] + 3^{1/6} \text{AiryBi}[x] \text{Gamma}\left[\frac{2}{3}\right] \right) \right\} \right\}$$

```
DSolve[{airyyht, y[0] == 0, y'[0] == 1}, y[x], x]
```

$$\left\{ \left\{ y[x] \rightarrow \frac{1}{6} \left(-3 3^{1/3} \text{AiryAi}[x] \text{Gamma}\left[\frac{1}{3}\right] + 3^{5/6} \text{AiryBi}[x] \text{Gamma}\left[\frac{1}{3}\right] \right) \right\} \right\}$$

Näiden kuvaajista on nähtävissä eräitä toisen kertaluvun homogeeniyhtälölle luonteenomaisia piirteitä:

```
Plot[{rtk1, rtk2}, {x, -15, 2}]
```



- Graphics -

Jos differentiaaliyhtälössä $y'' + k y = 0$ vakio k on positiivinen, kyseessä on vakiokertoiminen yhtälö, jonka ratkaisuna on $y = C_1 \sin(\sqrt{k} x) + C_2 \cos(\sqrt{k} x)$, ts. sini-kosini-värähtely. Värähtelyn taajuus on sitä suurempi, mitä suurempi k on. Negatiivisilla muuttujan x arvoilla Airyn yhtälö on tämäntyyppinen: Yhtälö voidaan kirjoittaa muotoon $y'' + |x| y = 0$, ja sen ratkaisuna näyttää olevan värähtely, jonka taajuus kasvaa, kun $|x|$ kasvaa.

Vastaavalla tavalla Airyn yhtälö voidaan rinnastaa positiivisilla muuttujan arvoilla yhtälöön $y'' - k y = 0$, missä k on positiivinen. Tämän ratkaisut muodostuvat eksponenttifunktioista: $y = C_1 e^{\sqrt{k} x} + C_2 e^{-\sqrt{k} x}$.

Kuvaajat näyttävät myös toisen kertaluvun homogeeniyhtälöiden ratkaisuille tyypillisen ominaisuuden: Jos kahdella lineaarisesti riippumattomalla ratkaisulla on nollakohtia, nämä vuorottelevat. Toisen ratkaisun kahden peräkkäisen nollakohtien välissä on täsmälleen yksi toisen ratkaisun nollakohta. Todistus perustuu Wronskin determinantin ominaisuuksiin.

Linkit

[lineaarisen ja homogeenisen yhtälön ratkaisujoukko](#)

[Wronskin determinantti](#)

[algebraalinen ratkaiseminen Mathematicalla](#) (symalg.nb)

[alkuehtoa vastaava yksittäisratkaisu Mathematicalla](#) (rtkalk.nb)

[Airyn yhtälön numeerinen ratkaiseminen](#) (numryh.nb)

[Airyn yhtälön sarjaratkaisu](#)

SKK 27.04.2001